Rules for integrands of the form $(g \sin[e+f x])^p (a+b \sec[e+f x])^m$

1: $\int (g \sin[e+f x])^p (a+b \sec[e+f x])^m \, dx$ when $m \in \mathbb{Z}$

- **Derivation: Algebraic normalization**
- **Basis: If** $m \in \mathbb{Z}$, **then** $(a+b \sec[z])^m == \frac{(b+a \cos[z])^m}{\cos[z]^m}$
- **Rule: If** $m \in \mathbb{Z}$, **then**

$$\int (g \sin[e+f x])^p (a+b \sec[e+f x])^m \, dx \longrightarrow \int \frac{(g \sin[e+f x])^p (b+a \cos[e+f x])^m}{\cos[e+f x]^m} \, dx$$

- **Program code:**

```
Int[(g_.*cos[e_.+f_.*x_])^p_.*(a_+b_.*csc[e_.+f_.*x_])^m_.,x_Symbol] :=
  Int[(g*Cos[e+f*x])^p*(b+a*Sin[e+f*x])^m/Sin[e+f*x]^m,x] /;
FreeQ[{a,b,e,f,g,p},x] && IntegerQ[m]
```

2. $\int \sin[e+f x]^p (a+b \sec[e+f x])^m \, dx$ when $\frac{p-1}{2} \in \mathbb{Z}$

1: $\int \sin[e+f x]^p (a+b \sec[e+f x])^m \, dx$ when $\frac{p-1}{2} \in \mathbb{Z} \bigwedge a^2-b^2 == 0$

- **Derivation: Integration by substitution**
- **Basis: If** $\frac{p-1}{2} \in \mathbb{Z} \bigwedge a^2-b^2 == 0$, **then** $\sin[e+f x]^p == \frac{1}{f b^{p-1}} \text{Subst}\left[\frac{(-a+b x)^{\frac{p-1}{2}} (a+b x)^{\frac{p-1}{2}}}{x^{p+1}}, x, \sec[e+f x]\right] \partial_x \sec[e+f x]$
- **Rule: If** $\frac{p-1}{2} \in \mathbb{Z} \bigwedge a^2-b^2 == 0$, **then**

$$\int \sin[e+f x]^p (a+b \sec[e+f x])^m \, dx \longrightarrow \frac{1}{f b^{p-1}} \text{Subst}\left[\int \frac{(-a+b x)^{\frac{p-1}{2}} (a+b x)^{m+\frac{p-1}{2}}}{x^{p+1}} \, dx, x, \sec[e+f x]\right]$$

- **Program code:**

```
Int[cos[e_.+f_.*x_]^p_.*(a_+b_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  -1/(f*b^(p-1))*Subst[Int[(-a+b*x)^((p-1)/2)*(a+b*x)^(m+(p-1)/2)/x^(p+1),x],x,Csc[e+f*x]] /;
FreeQ[{a,b,e,f,m},x] && IntegerQ[(p-1)/2] && EqQ[a^2-b^2,0]
```

2: $\int \text{Sin}[e+f\,x]^p\,(a+b\,\text{Sec}[e+f\,x])^m\,dx$ when $\frac{p-1}{2}\in\mathbb{Z}\,\bigwedge\,a^2-b^2\neq 0$

Derivation: Integration by substitution

- **Basis: If** $\frac{p-1}{2}\in\mathbb{Z}$**, then** $\text{Sin}[e+f\,x]^p == \frac{1}{f}\,\text{Subst}\Big[\frac{(-1+x)^{\frac{p-1}{2}}\,(1+x)^{\frac{p-1}{2}}}{x^{p+1}},\ x,\ \text{Sec}[e+f\,x]\Big]\,\partial_x\text{Sec}[e+f\,x]$
- **Rule: If** $\frac{p-1}{2}\in\mathbb{Z}\,\bigwedge\,a^2-b^2\neq 0$**, then**

$$\int \text{Sin}[e+f\,x]^p\,(a+b\,\text{Sec}[e+f\,x])^m\,dx \longrightarrow \frac{1}{f}\,\text{Subst}\Big[\int\frac{(-1+x)^{\frac{p-1}{2}}\,(1+x)^{\frac{p-1}{2}}\,(a+b\,x)^m}{x^{p+1}}\,dx,\ x,\ \text{Sec}[e+f\,x]\Big]$$

Program code:

```
Int[cos[e_.+f_.*x_]^p_.*(a_+b_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  -1/f*Subst[Int[(-1+x)^((p-1)/2)*(1+x)^((p-1)/2)*(a+b*x)^m/x^(p+1),x],x,Csc[e+f*x]] /;
FreeQ[{a,b,e,f,m},x] && IntegerQ[(p-1)/2] && NeQ[a^2-b^2,0]
```

3: $\int \frac{(a+b\,\text{sec}[e+f\,x])^m}{\text{Sin}[e+f\,x]^2}\,dx$

Derivation: Integration by parts

- **Basis:** $\int\frac{1}{\text{Sin}[e+f\,x]^2}\,dx == -\frac{\text{Cot}[e+f\,x]}{f}$
- **Basis:** $-\frac{\text{Cot}[e+f\,x]}{f}\,\partial_x\,(a+b\,\text{Sec}[e+f\,x])^m == -b\,m\,\text{Sec}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m-1}$

Rule:

$$\int\frac{(a+b\,\text{sec}[e+f\,x])^m}{\text{Sin}[e+f\,x]^2}\,dx \longrightarrow -\frac{\text{Cot}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^m}{f}+b\,m\int\text{Sec}[e+f\,x]\,(a+b\,\text{Sec}[e+f\,x])^{m-1}\,dx$$

Program code:

```
Int[(a_+b_.*csc[e_.+f_.*x_])^m_/cos[e_.+f_.*x_]^2,x_Symbol] :=
  Tan[e+f*x]*(a+b*Csc[e+f*x])^m/f + b*m*Int[Csc[e+f*x]*(a+b*Csc[e+f*x])^(m-1),x] /;
FreeQ[{a,b,e,f,m},x]
```

4: $\int (g \operatorname{Sin}[e + f x])^p (a + b \operatorname{Sec}[e + f x])^m \, dx$ when $a^2 - b^2 == 0 \lor (2 m \mid p) \in \mathbb{Z}$

Derivation: Piecewise constant extraction

- **Basis:** $\partial_x \frac{\operatorname{Cos}[e+f x]^m (a+b \operatorname{Sec}[e+f x])^m}{(b+a \operatorname{Cos}[e+f x])^m} == 0$
- **Rule: If** $a^2 - b^2 == 0 \lor (2 m \mid p) \in \mathbb{Z}$, **then**

$$\int (g \operatorname{Sin}[e + f x])^p (a + b \operatorname{Sec}[e + f x])^m \, dx \longrightarrow \frac{\operatorname{Cos}[e + f x]^m (a + b \operatorname{Sec}[e + f x])^m}{(b + a \operatorname{Cos}[e + f x])^m} \int \frac{(g \operatorname{Sin}[e + f x])^p (b + a \operatorname{Cos}[e + f x])^m}{\operatorname{Cos}[e + f x]^m} \, dx$$

- **Program code:**

```
Int[(g_.*cos[e_.+f_.*x_])^p_.*(a_+b_.*csc[e_.+f_.*x_])^m_,x_Symbol] :=
  Sin[e+f*x]^FracPart[m]*(a+b*Csc[e+f*x])^FracPart[m]/(b+a*Sin[e+f*x])^FracPart[m]*
    Int[(g*Cos[e+f*x])^p*(b+a*Sin[e+f*x])^m/Sin[e+f*x]^m,x] /;
FreeQ[{a,b,e,f,g,m,p},x] && (EqQ[a^2-b^2,0] || IntegersQ[2*m,p])
```

X: $\int (g \operatorname{Sin}[e + f x])^p (a + b \operatorname{Sec}[e + f x])^m \, dx$

Rule:

$$\int (g \operatorname{Sin}[e + f x])^p (a + b \operatorname{Sec}[e + f x])^m \, dx \longrightarrow \int (g \operatorname{Sin}[e + f x])^p (a + b \operatorname{Sec}[e + f x])^m \, dx$$

- **Program code:**

```
Int[(g_.*cos[e_.+f_.*x_])^p_.*(a_+b_.*csc[e_.+f_.*x_])^m_.,x_Symbol] :=
  Unintegrable[(g*Cos[e+f*x])^p*(a+b*Csc[e+f*x])^m,x] /;
FreeQ[{a,b,e,f,g,m,p},x]
```

Rules for integrands of the form $(g\,Csc[e+f\,x])^p\,(a+b\,Sec[e+f\,x])^m$

x: $\int (g\,Csc[e+f\,x])^p\,(a+b\,Sec[e+f\,x])^m\,dx$ when $p\notin\mathbb{Z}\,\land\,m\in\mathbb{Z}$

- **Derivation: Algebraic normalization**
- **Basis: If** $m\in\mathbb{Z}$, **then** $(a+b\,Sec[z])^m == \frac{(b+a\,Cos[z])^m}{Cos[z]^m}$
- **Rule: If** $p\notin\mathbb{Z}\,\land\,m\in\mathbb{Z}$, **then**

$$\int (g\,Csc[e+f\,x])^p\,(a+b\,Sec[e+f\,x])^m\,dx\,\longrightarrow\,\int \frac{(g\,Csc[e+f\,x])^p\,(b+a\,Cos[e+f\,x])^m}{Cos[e+f\,x]^m}\,dx$$

- **Program code:**

```
(* Int[(g_.*sec[e_.+f_.*x_])^p_*(a_+b_.*csc[e_.+f_.*x_])^m_.,x_Symbol] :=
  Int[(g*Sec[e+f*x])^p*(b+a*Sin[e+f*x])^m/Sin[e+f*x]^m,x] /;
FreeQ[{a,b,e,f,g,p},x] && Not[IntegerQ[p]] && IntegerQ[m] *)
```

1: $\int (g\,Csc[e+f\,x])^p\,(a+b\,Sec[e+f\,x])^m\,dx$ when $p\notin\mathbb{Z}$

- **Derivation: Piecewise constant extraction**
- **Basis:** $\partial_x\left((g\,Csc[e+f\,x])^p\,Sin[e+f\,x]^p\right) == 0$
- **Rule: If** $p\notin\mathbb{Z}$, **then**

$$\int (g\,Csc[e+f\,x])^p\,(a+b\,Sec[e+f\,x])^m\,dx\,\longrightarrow\,g^{IntPart[p]}\,(g\,Csc[e+f\,x])^{FracPart[p]}\,Sin[e+f\,x]^{FracPart[p]}\int \frac{(a+b\,Sec[e+f\,x])^m}{Sin[e+f\,x]^p}\,dx$$

- **Program code:**

```
Int[(g_.*sec[e_.+f_.*x_])^p_*(a_+b_.*csc[e_.+f_.*x_])^m_.,x_Symbol] :=
  g^IntPart[p]*(g*Sec[e+f*x])^FracPart[p]*Cos[e+f*x]^FracPart[p]*Int[(a+b*Csc[e+f*x])^m/Cos[e+f*x]^p,x] /;
FreeQ[{a,b,e,f,g,m,p},x] && Not[IntegerQ[p]]
```